
Abstract

The shapes macros for METAPOST provide regular polygons, their corre-
sponding reentrant stars, and images demonstrating fractions. These macros
are quite configurable.

1 Introduction

The shapes macros are not revolutionary, and in fact are quite simple; however,
I spent some time generalizing them for a text I’m currently working on, and so I
thought they might be useful for the general populace. They are divided into two
main groups: regular polygons and their corresponding reentrant star shapes; and
fractionals, circles divided into a certain number of parts with the desired fraction
filled in.

This document was typeset in accordance with the docstrip utility, which
allows the automatic extraction of code and documentation from the same docu-
ment.

2 Prerequisites and Conventions

Some prerequesites for using this package are METAPOST itself (obviously). If
you’re using the package with LATEX, the gmp package would probably be help-

ful; be sure to use the latex package option. These should be packaged in any
reasonably modern LATEX system, such as TEXLive or MikTEX.

This documentation assumes nothing about your personal TEX or METAPOST

environment. ConTEXt and the various forms of LuaTEX have METAPOST built-
in; with pdfLATEX, the author’s choice, one can use the gmp package to include the
source directly in one’s document (that’s what’s been done in this documentation)
or develop a simple script to compile them afterwards and include them in the
source via \includegraphics (probably the quickest option, since compilation is
done in advance). Here, we simply post the plain vanilla METAPOST code, and
let you work out those details however you prefer.

3 The Shapes Macros

We begin with the simple shapes macros, which are about as basic as they can be.
All of these shapes default to circles with a one inch diameter, so you can

scaled them with that kept in mind.
The simplest case is with a regular polygon, like so:

modfig(6,0);

draw modfigure;

Of course, these drawn figures can be manipulated in the usual ways:

modfig(6,0);

draw modfigure rotated (45) withcolor red;

The two arguments to modfig are simple: the first tells the number of sides
desired, while the second means no numbers are printed if 0, and print numbers
if 1:

modfig(8,1);

draw modfigure;

If you don’t want the circle to be printed, simply tell METAPOST with
modcircle := false; and it will not print it:

modcircle := false;

modfig(8,1);

draw modfigure;

modcircle defaults to true.
Notice that modfig doesn’t care if there’s a circle or not; if you want numbers

on the vertices, it will print them there.
You can adjust the width of the lines by specifying modcirclepen and

modshapepen:

modcirclepen := pencircle scaled 3;

modshapepen := pencircle scaled 2;

modfig(4,0);

draw modfigure;

Both modcirclepen and modshapepen default to pencircle scaled 1.
We can do essentially the same thing with reentrant star figures with modstar.

Unlike modfig, modstar takes three arguments: the number of vertices, whether
or not you want those vertices numbered, and how many points you want to skip
as you go around the circle.

modstar(11,1,3);

draw modfigure;

Note that the third argument actually skips n−1 points, not n points. But all
the same parameters we saw when looking at modfig will still work in the same
way:

modcirclepen := pencircle scaled 1;

modshapepen := pencircle scaled 2;

modstar(11,1,4);

draw modfigure;

And that’s about all there is to it.

4 Fraction Images

The following macros are useful for demonstrating the nature and size of frac-
tions in a visible way. The name of the game here is fraccirc, which takes two
arguments: the number of parts to be filled, and the number of parts in the whole:

fraccirc(7)(12);

draw thefrac;

Note that, due to the internal implementation, these two arguments must each
be enclosed in their own parentheses.

We can adjust these things as appropriate. For example, we can fill with blue
rather than red, and use thicker lines:

fracfillcolor := blue;

fraccirclepen := pencircle scaled 1;

fractionpen := pencircle scaled 2;

fraccirc(4)(7);

draw thefrac;

As the above example suggests, fracfillcolor gives the color with which the
portion of the fraction should be filled; it defaults to red. fraccirclepen is the
pen used to draw the circle around the fraction; it defaults to pencircle scaled 1.5.
Finally, fractionpen is the pen used to draw the partition and the circle imme-
diately surrounding them; it defaults to pencircle scaled 1.

5 Implementation

1 color fracfillcolor; fracfillcolor := red;

2 pen fraccirclepen; fraccirclepen := pencircle scaled 1.5;

3 pen fractionpen; fractionpen := pencircle scaled 1;

4 def fraccirc(suffix x)(expr y) =

5 radius := 1in;

6 ticklen := radius/24;

7 path circ; circ := fullcircle scaled radius;

8 pair p[]; pair q[]; pair r[]; pair s[];

9 p[0] := (0,0) shifted (0,radius/2);

10 q[0] := p[0] shifted (0,ticklen);

11 r[0] := p[0] shifted (0,-ticklen);

12 s[0] := r[0];

13 picture thefrac;

14 picture addition;

15 thefrac := image(pickup fraccirclepen; draw circ;);

16 addition := image(pickup fractionpen; draw p[0]--q[0];);

17 addto thefrac also addition;

18 for i=1 upto 12:

19 p[i] := p[i-1] rotatedaround ((0,0),-30);

20 q[i] := q[i-1] rotatedaround ((0,0),-30);

21 addition := image(pickup fractionpen; draw p[i]--q[i];);

22 addto thefrac also addition;

23 endfor;

24 addition := image(pickup fractionpen; draw (0,0)--r[0];);

25 addto thefrac also addition;

26 pair t; pair q;

27 addition := image(

28 t = r[0] rotatedaround ((0,0),-((360/y)*x)/2);

29 q = r[0] rotatedaround ((0,0),-((360/y)*x));

30 fill r[0]--(0,0)--q..t..cycle withcolor fracfillcolor;

31 draw r[0]--(0,0)--q..t..cycle;

32 );

33 addto thefrac also addition;

34 for i=1 upto y:

35 r[i] := r[i-1] rotatedaround ((0,0),-(360/y));

36 s[i] := r[i-1] rotatedaround ((0,0),-(360/y)/2);

37 addition := image(%

38 pickup fractionpen;%

39 draw (0,0)--r[i];

40 draw r[i-1]..s[i]..r[i];

41 );

42 addto thefrac also addition;

43 endfor;

44 enddef;

45 % put in the pens for the modular shapes

46 boolean modcircle; modcircle := true;

47 pen modcirclepen; modcirclepen := pencircle scaled 1;

48 pen modshapepen; modshapepen := pencircle scaled 1;

49 def modstar(expr numpoints,numbers,numstar) =

50 picture modfigure;

51 if (modcircle = true):

52 modfigure := image(draw fullcircle scaled 1in withpen modcirclepen;);

53 else:

54 modfigure := image();

55 fi;

56 pickup modshapepen;

57 pair p[]; pair q[];

58 picture addition;

59 p[0] = (0,0.5in);

60 q[0] = (0,0.6in);

61 if (numbers = 1):

62 addition := image(label("0",q[0]));

63 addto modfigure also addition;

64 fi

65 for i=1 upto numpoints:

66 p[i] = p[i-1] rotatedaround ((0,0),-(360/numpoints));

67 q[i] = q[i-1] rotatedaround ((0,0),-(360/numpoints));

68 if (numbers = 1):

69 if (i <> numpoints):

70 addition := image(label(decimal i,q[i]));

71 addto modfigure also addition;

72 fi

73 fi

74 endfor

75 for i=0 upto numpoints:

76 if (i < numstar):

77 addition := image(%

78 draw p[i]--p[i+numstar];

79 draw p[i]--p[numpoints - numstar + i];

80 );

6



81 addto modfigure also addition;

82 elseif (i >= (numstar*2)):

83 addition := image(draw p[i-numstar]--p[i]);

84 addto modfigure also addition;

85 fi

86 endfor

87 addto modfigure also addition;

88 enddef;

89 def modfig(expr numpoints,numbers) =

90 picture modfigure;

91 if (modcircle = true):

92 modfigure := image(draw fullcircle scaled 1in withpen modcirclepen);

93 else:

94 modfigure := image();

95 fi

96 pair p[]; pair q[];

97 picture addition;

98 pickup modshapepen;

99 p[0] = (0,0.5in);

100 q[0] = (0,0.6in);

101 if (numbers = 1):

102 addition := image(label("0",q[0]));

103 addto modfigure also addition;

104 fi

105 for i=1 upto numpoints:

106 p[i] = p[i-1] rotatedaround ((0,0),-(360/numpoints));

107 q[i] = q[i-1] rotatedaround ((0,0),-(360/numpoints));

108 addition := image(draw p[i-1]--p[i]);

109 addto modfigure also addition;

110 if (numbers = 1):

111 if (i <> numpoints):

112 addition := image(label(decimal i,q[i]));

113 addto modfigure also addition;

114 fi

115 fi

116 endfor;

117 enddef;

